
RESUMEN. Introduciremos de manera amigable las nociones elementales de la geometría
riemanniana (conexión de Levi-Civita, geodésicas, curvatura, isometrías, etc.). Para ejem-
plificar estos conceptos estudiaremos métricas invariantes a izquierda en grupos de Lie
tridimensionales. Trataremos de mostrar con estos ejemplos cómo podemos ayudarnos de
SageMath para resolver problemas de geometría riemanniana.

1. INTRODUCCIÓN

La idea de este curso es presentar, de manera informal, algunos de los conceptos básicos
que se estudian en geometría riemanniana homogénea (esto incluye grupos y álgebras de
Lie). A la par iremos viendo cómo se implementan estos conceptos en el software SageMath
y algunas de sus funcionalidades.

En SageMath las variedades diferenciables están implementadas en el proyecto Sage-
Manifolds. Al momento de dictar el curso, la última versión disponible de SageMath era
la versión 9.3. Sin embargo, al momento de terminar estas notas ya se cuenta con la ver-
sión 9.4. Estas notas están, dentro de lo posible, adaptadas a la última versión de Sage-
Math. Las notebooks originales (así como algunas imágenes ilustrativas que no tienen la
calidad suficiente como para ser incluidas en estas actas) se encuentran en el repositorio
https://github.com/silvioreggiani/curso-monteiro.

[1]: version()

[1]: 'SageMath version 9.4, Release Date: 2021-08-22'

Algunos enlaces relevantes para quienes quieran ejecutar las notebooks mencionadas más
arriba son:

https://doc.sagemath.org/html/en/installation/ para la instalación de SageMath y Jupy-
ter.
Alternativamente, las notebooks pueden ejecutarse en https://cocalc.com/ con una
cuenta gratuita (la versión gratuita es un poco lenta, pero más que suficiente para
este curso).
Para quienes no estén familiarizados con SageMath, se recomienda hacer antes algún
tutorial, por ejemplo https://doc.sagemath.org/html/en/tutorial/index.html
Finalmente, https://sagemanifolds.obspm.fr/ es la página oficial del proyecto Sage-
Manifolds. Además de la documentación oficial, allí se pueden encontrar los cambios
hechos en las últimas actualizaciones.

Con respecto a la parte matemática del curso, la exposición será autocontenida pero infor-
mal. Un curso de calculo en varias variables o geometría de curvas y superficies alcanzará
como base. Para profundizar en los temas que cubriremos se recomienda estudiar los con-
ceptos elementales de variedades diferenciables y riemannianas, esto puede encontrarse, por
ejemplo, en los primeros capítulos del libro de do Carmo [dC92].

2. VARIEDADES DIFERENCIABLES

Decimos que un espacio topológico M es una variedad diferenciable de dimensión n si
tenemos coordenadas locales x1, . . . ,xn en (un entorno abierto de) cada punto p ∈ M;
el cambio de coordenadas (x1, . . . ,xn) 7→ (y1, . . . ,yn) es diferenciable (para nosotros
diferenciablilidad significará de clase C∞);
la topología de M es suficientemente buena (espacio localmente euclídeo).

Para definir una variedad diferenciable en Sage tenemos el comando Manifold(). Por
ejemplo, una variedad diferenciable M de dimensión 3 la indicamos como sigue.

[2]: M = Manifold(3, 'M'); M

[2]: 3-dimensional differentiable manifold M

Podemos pedirle a Sage que nos muestre los resultados usando LATEX. Sin embargo, a
veces el output mostrado de esta manera es demasiado largo para ser visualizado correcta-
mente. Es por eso que para hacer la exposición más amena, alteremos entre el modo LATEXy
el modo código para mostrar los resultados.

[3]: %display latex

M

[3]:

M

Investiguemos un poco qué es lo que acabamos de definir.

[4]: %display plain #mostramos los resultados en modo código

M.parent()

[4]: <class 'sage.manifolds.differentiable.manifold.

DifferentiableManifold_with_category'>

Lo anterior nos dice, rápidamente hablando, que M es un objeto de la clase de las varieda-
des diferenciables. No nos preocuparemos mucho por saber qué es lo que significa un objeto
y una clase. Solo diremos que esto viene heredado de Python, que es el software en el que
SageMath está escrito. Otra manera, más matemática, de entender qué es M para Sage, es
usando el método .category().

[5]: %display latex #volvemos a mostrar el output con LaTeX

M.category()

[5]:

SmoothR

Un mismo objeto puede estar en varias categorías.

[6]: %display plain

M.categories()

[6]: [Category of smooth manifolds over Real Field with 53 bits of

precision,

Category of differentiable manifolds over Real Field with 53 bits

of precision,

Category of manifolds over Real Field with 53 bits of precision,

Category of topological spaces,

Category of sets,

Category of sets with partial maps,

Category of objects]

Observemos que todavía no tenemos definido ningún sistema de coordenadas en M, y
por ende la lista

[7]: %display latex

M.atlas()

[7]:

[ ]

es una lista vacía. Aquí, el método .atlas() busca los sistemas de coordenadas definidos
por el usuario y no hay que confundirlo con el atlas maximal de una variedad diferenciable.
Definamos una carta en M.

[8]: x.<x1,x2,x3> = M.chart()

Con este comando hemos definido una carta global x : M → R3 cuyas funciones coorde-
nadas son x1,x2,x3. (Una explicación un poco más técnica es que .chart() es un método
de la clase Manifold que usamos para definir los sistemas de coordenadas.)

Ahora vemos que el atlas de M ya no está vacío.

[9]: M.atlas()

[9]:

[(M,(x1,x2,x3))]

Podemos definir un punto p∈M especificando sus coordenadas en la carta que acabamos
de definir y podemos mostrar p de diferentes maneras.

[10]: p = M.point((1,-7,12), name='p')

print(p)

display(p)

display(p in M)

p.coord()

Point p on the 3-dimensional differentiable manifold M

p

True

[10]:

(1,−7,12)

La variedad que acabamos de definir no es otra cosa que R3 ya que las coordenadas se
mueven entre −∞ y +∞.

[11]: x.coord_range()

[11]:

x1 : (−∞,+∞) ; x2 : (−∞,+∞) ; x3 : (−∞,+∞)

En realidad, SageManifolds viene con un comando EuclideanSpace() que nos permite
definir las variedades Rn de manera más sencilla.

[12]: E3 = EuclideanSpace(3)

display(E3)

E3.atlas()

E3

[12]: [(
E3,(x,y,z)

)]
Notar, sin embargo, que para Sage, M y E3 son dos variedades distintas.

[13]: M == E3

[13]:

False

De hecho, los espacios euclídeos vienen con estructuras adicionales, como por ejemplo
la métrica riemanniana usual (inducida por el producto escalar). Observemos que el plano
E2 viene también con coordenadas polares.

[14]: E2 = EuclideanSpace(2)

E2.atlas()

[14]: [(
E2,(x,y)

)]
Pero no vienen cargadas por defecto, tenemos que hacerlo manualmente.

[15]: coord_cart = E2.cartesian_coordinates()

coord_polar = E2.polar_coordinates()

E2.atlas()

[15]: [(
E2,(x,y)

)
,
(
E2,(r,φ)

)]
Naturalmente, (r,φ) ∈ R+× [0,2π).

[16]: coord_polar.coord_range()

[16]:

r : (0,+∞) ; φ : [0,2π] (periodic)

Definamos un punto genérico en el plano y veamos el cambio de coordenadas.

[17]: p = E2.point(coord_cart[:])

display(p.coord())

p.coord(coord_polar)

(x,y)

[17]: (√
x2 + y2,arctan(y,x)

)
Observar que el método .point() usa la carta por defecto, que en nuestro caso son las

coordenadas cartesianas.
[18]: E2.default_chart()

[18]: (
E2,(x,y)

)

Pero también podríamos definir un punto genérico en coordenadas polares. Como ya
vimos antes el método .coord() también usa la carta por defecto, pero podemos pedirle
que nos muestre las coordenadas con respecto a cualquier carta.

[19]: q = E2.point(coord_polar[:],chart=coord_polar)

display(q.coord())

q.coord(coord_polar)

(r cos(φ) ,r sin(φ))

[19]:

(r,φ)

Es posible graficar como se ve una carta coordenada con respecto a la otra.

[20]: coord_polar.plot(coord_cart,color={coord_polar[0]:'blue',

coord_polar[1]:'red'}

)

[20]:

[21]: coord_cart.plot(coord_polar,color={coord_cart[0]:'blue',

coord_cart[1]:'red'}

)

[21]:

Es interesante saber que el método chart1.plot(chart2) funciona en variedades de
cualquier dimensión (obviamente, eligiendo secciones de dimensión 2 o 3 para mostrar los
gráficos).

3. FUNCIONES DIFERENCIABLES

Si N y M son dos variedades diferenciables, una función f : N → M se dice diferenciable
si lo es cuando la escribimos en coordenadas locales. Es decir, si x1, . . . ,xn son coordenadas
locales alrededor de p ∈ N e y1, . . . ,ym son coordenadas locales alrededor de f (p) ∈ M,
podemos pensar localmente a f como una función Rn → Rm (y esta función es la que debe
ser diferenciable).

Definamos la función d : R2 → R que calcula el cuadrado de la distancia al origen. De
paso, mostramos una forma distinta de definir la variedad diferenciable R.

[22]: R.<t> = manifolds.RealLine(); print(R)

Real number line R
En la versión original de las notebooks usábamos directamente el comando RealLine().

En la nueva versión esto ya es obsoleto por lo que ahora debemos definir la recta usando
manifolds.RealLine().

[23]: E2.<x,y> = EuclideanSpace(2); E2

[23]:

E2

Hemos vuelto a definir el plano, esta vez indicando el nombre de las coordenadas, lo cual
nos resulta más cómodo para definir la función d. Para definir una función diferenciable de
M en N podemos usar M.diff_map(N, ...).

[24]: d = E2.diff_map(R, x^2+y^2)

d.display()

[24]:

E2 −→ R
(x,y) 7−→ t = x2 + y2

Luego d es un elemento de lo que usualmente se denota por C∞(R2,R). En el lenguaje
de categorías SageMath presenta esto de la siguiente forma:

[25]: display(d.parent())

print(d.parent()) #verbose

Hom
(
E2,R

)
Set of Morphisms from Euclidean plane E�2 to Real number line R in

Category of smooth connected manifolds over Real Field with 53 bits

of precision

Las funciones se pueden evaluar en puntos como lo haríamos normalmente.

[26]: p = E2.point((-1,2))

d(p).coord()

[26]:

(5)

Los ejemplos hasta ahora son bastante simples porque las variedades que definimos son
también sencillas. En general, para cubrir todos los puntos de una variedad diferenciable M
hace falta más de una carta. En estos casos, para definir una función diferenciable f usando
diff_map(), tendremos que pasar como argumentos las expresiones en coordenadas de f
con respecto a un subconjunto de cartas que cubra toda la variedad M. En breve veremos
ejemplos de esto también.

4. CURVAS Y VECTORES TANGENTES

Una curva en M es una función diferenciable

c : I → M

en donde I es un intervalo de R. La velocidad de la curva c(t), con respecto a una carta
x1, . . . ,xn se define como

c′(t) 7→ (c′1(t), . . . ,c
′
n(t))

en donde c1(t), . . . ,cn(t) son las coordenadas de c(t) en la carta antes mencionada.
Es posible definir la velocidad de una curva independientemente del sistema de coorde-

nadas, considerando la clase de equivalencia de estos vectores en Rn si se puede pasar de
uno a otro vía (la matriz jacobiana de) un cambio de coordenadas. Alternativamente, esto se
puede hacer considerando a los vectores tangentes como derivaciones (aplicaciones lineales
que cumplen la regla de Leibniz) del álgebra de funciones diferenciables de M.

A modo de ejemplo una espiral de Fermat en el plano. Recordemos que en coordenadas
polares una espiral de Fermat, que empieza en el origen, tiene la forma

r =±a
√

φ , φ ≥ 0

[27]: R.<t> = manifolds.RealLine()

E2 = EuclideanSpace(2)

coord_cart.<x,y> = E2.cartesian_coordinates()

coord_polar.<r,ph> = E2.polar_coordinates()

Algunas aclaraciones sobre lo que sigue. En lugar de definir la curva usando el método
R.diff_map(E2, ...) es preferible hacerlo con el método E2.curve(...). Igualmen-
te definimos la recta R.<t> = manifolds.RealLine() para que Sage cree la variable
simbólica t. Otra forma de hacer esto sería con t = var('t'). Ahora sí podemos definir
nuestra curva.

[28]: c = E2.curve((t,t^2), (t,0,3),chart=coord_polar)

print(c)

c.display()

Curve in the Euclidean plane E�2

[28]:
(0,3) −→ E2

t 7−→ (x,y) =
(
t cos

(
t2
)
, t sin

(
t2
))

t 7−→ (r,φ) =
(
t, t2
)

Observemos que SageMath nos muestra la expresión de c(t) en los dos sistemas de coor-
denadas disponibles. Grafiquemos la espiral de Fermat, primero en coordenadas polares:

[29]: c.plot(coord_polar)

[29]:

Y luego en coordenadas cartesianas:

[30]: c1 = E2.curve(c.coord_expr(coord_cart), (t,0,3))

g1 = c1.plot(thickness=2)

show(g1)

Finalmente definamos y grafiquemos la velocidad de esta curva.

[31]: v = c1.tangent_vector_field()

print(v)

display(v.display())

g2 = v.plot(chart=coord_cart, number_values=10,scale=0.1)

show(g1+g2)

Vector field along the Real interval (0, 3) with values on the

Euclidean plane E�2

(
−2 t2 sin

(
t2)+ cos

(
t2))ex +

(
2 t2 cos

(
t2)+ sin

(
t2))ey

5. ESPACIO TANGENTE

Si p ∈ M, el espacio tangente a M en p se define por

TpM = {v = c′(0) : c curva en M tal que c(0) = p}.

Dado v ∈ TpM, podemos calcular la derivada direccional de una función f : M → R en
la dirección de v por:

v( f ) =
d
dt

∣∣∣∣
0

f (c(t))

en donde c(t) es una curva en M tal que c(0) = p y c′(0) = v. Esta definición no depende
de la curva elegida (siempre que tenga velocidad v). Más aún, las derivadas direccionales
son independientes de las coordenadas locales. Esto último es muy importante, porque nos
dice que el espacio vectorial TpM existe en cada punto y se puede “construir” independien-
temente de las coordenadas locales.

Veamos cómo se trabaja con los espacios tangentes en Sage. Primero definimos un punto
en una variedad de dimensión 4.

[32]: M = Manifold(4, 'M')

M_chart.<x,y,z,w> = M.chart()

p = M.point((1,0,-5,1), name='p')

Y usamos el método .tangent_space() para acceder a TpM. Observemos que debemos
pasar como argumento el punto base.

[33]: TpM = M.tangent_space(p)

print(TpM)

TpM

Tangent space at Point p on the 4-dimensional differentiable

manifold M

[33]:

Tp M

Como TpM es un espacio vectorial, está en muchas categorías:

[34]: %display plain

TpM.categories()

[34]: [Category of finite dimensional vector spaces over Symbolic Ring,

Category of finite dimensional modules over Symbolic Ring,

Category of vector spaces over Symbolic Ring,

Category of modules over Symbolic Ring,

Category of bimodules over Symbolic Ring on the left and Symbolic

Ring on the right,

Category of right modules over Symbolic Ring,

Category of left modules over Symbolic Ring,

Category of commutative additive groups,

Category of additive groups,

Category of additive inverse additive unital additive magmas,

Category of commutative additive monoids,

Category of additive monoids,

Category of additive unital additive magmas,

Category of commutative additive semigroups,

Category of additive commutative additive magmas,

Category of additive semigroups,

Category of additive magmas,

Category of sets,

Category of sets with partial maps,

Category of objects]

Hay varios métodos interesantes para un espacio vectorial en SageMath. Por ejemplo,

[35]: %display latex

display(TpM.dim())

display(TpM.default_basis())

vp = TpM.an_element(); vp.display()

4

(
∂

∂x
,

∂

∂y
,

∂

∂ z
,

∂

∂w

)
[35]:

∂

∂x
+2

∂

∂y
+3

∂

∂ z
+4

∂

∂w

En la notación clásica el vector vp debería escribirse como

∂

∂x

∣∣∣∣
p
+2

∂

∂y

∣∣∣∣
p
+3

∂

∂ z

∣∣∣∣
p
+4

∂

∂w

∣∣∣∣
p
.

Aquí esto no es necesario, ya que al ser vp un elemento de TpM, SageMath puede inferir
que p es el punto base. Solamente deberemos tener un poco de cuidado de no confundirnos.
Definamos otro vector tangente

[36]: wp = TpM._element_constructor_([5,-2,1,0])

wp.display()

[36]:

5
∂

∂x
−2

∂

∂y
+

∂

∂ z
y veamos que efectivamente los vectores tangentes se pueden sumar.

[37]: display(vp + wp)

(vp + wp).display()

Tangent vector at Point p on the 4-dimensional differentiable manifold M
[37]:

6
∂

∂x
+4

∂

∂ z
+4

∂

∂w
Ahora veamos cómo calcular las derivadas direccionales. Técnicamente hablando, en

Sage podemos derivar, a lo largo de un vector tangente, campos escalares definidos sobre
M, en lugar de funciones M → R. Desde el punto de vista matemático, estos dos conceptos
coinciden, pero para Sage M.diff_map(R,...) y M.scalar_field() son cosas distintas.

[38]: R.<t> = manifolds.RealLine()

f = M.scalar_field(4*x*y + z^2 - cos(w*y))

f.display()

[38]:
M −→ R
(x,y,z,w) 7−→ 4xy+ z2 − cos(wy)

Para calcular la derivada direccional simplemente aplicamos el vector v al campo esca-
lar f .

[39]: vp(f)

[39]:

−22
Ahora veamos que v( f ) coincide con d

dt

∣∣
0 f (c(t)) para una curva con c(0) = p y c′(0) = v.

Primero recordemos quiénes eran p y v.

[40]: display(p.coord())

vp.display()

(1,0,−5,1)
[40]:

∂

∂x
+2

∂

∂y
+3

∂

∂ z
+4

∂

∂w
Luego, necesitamos una curva que en el origen pase por (1,0,−5,1) con velocidad

(1,2,3,4).

[41]: c = M.curve((1 + t, 2*t , -5 + 3*t, 1 + 4*t), t)

print(c)

c.display()

Curve in the 4-dimensional differentiable manifold M

[41]:
R −→ M
t 7−→ (x,y,z,w) = (t +1,2 t,3 t −5,4 t +1)

Efectivamente
[42]: c(0) == p

[42]:

True

Observemos que para calcular c′(0) tenemos que evaluar el campo de velocidades en el
punto R(0) (no lo podemos evaluar directamente en 0).

[43]: vc = c.tangent_vector_field()

vc.at(R(0)) == vp

[43]:

True

Finalmente, hacemos

[44]: %display plain

f(c(t))

[44]: 2*cos(t)�2*sin(4*t�2)�2 + 4*cos(4*t�2)*cos(t)*sin(4*t�2)*sin(t)

+ 2*cos(4*t�2)�2*sin(t)�2 + 17*t�2 - 22*t + 24

[45]: f(c(t)).diff(t) #derivada con respecto a t

[45]: 32*t*cos(4*t�2)*cos(t)�2*sin(4*t�2) + 32*t*cos(4*t�2)�2*cos(t)*sin(t)

- 32*t*cos(t)*sin(4*t�2)�2*sin(t)

- 32*t*cos(4*t�2)*sin(4*t�2)*sin(t)�2

+ 4*cos(4*t�2)*cos(t)�2*sin(4*t�2) + 4*cos(4*t�2)�2*cos(t)*sin(t)

- 4*cos(t)*sin(4*t�2)�2*sin(t) - 4*cos(4*t�2)*sin(4*t�2)*sin(t)�2

+ 34*t - 22

[46]: f(c(t)).diff(t).subs(t=0) #evaluada en t = 0

[46]: -22

6. DIFERENCIAL DE UNA FUNCIÓN

Al igual que en análisis en varias variables, la diferencial de una función diferenciable
f : M → N en el punto p ∈ M es la transformación lineal

d f |p : TpM → Tf (p)N

que mejor aproxima a f . Podemos definir la diferencial usando curvas: si v = c′(0) ∈ TpM,
entonces

d f |p(v) =
d
dt

∣∣∣∣
0

f (c(t))

También podemos trabajar en coordenadas: si x1, . . . ,xm son coordenadas locales en p ∈
M; y1, . . . ,ym son coordenadas locales en f (p) ∈ N y f1, . . . , fn es la expresión de f con
respecto a este par de cartas, entonces la matriz de d f |p con respecto a las bases ∂

∂xi

∣∣
p y

∂

∂y j

∣∣
f (p) es la matriz jacobiana (

∂ fi

∂x j

∣∣∣∣
p

)
i, j

Sage nos permite calcular la diferencial de una función en un punto dado usando el mé-
todo f.differential(p). A modo de ejemplo calculemos la diferencial del cambio de
coordenadas polares a cartesianas en el plano.

[47]: %display latex

R2_polar = Manifold(2, r'\mathbb R^2_\text{polar}')

polar_range = r'r:(0,+oo) theta:[0,2*pi]:periodic'

R2_polar_chart.<r,theta> = R2_polar.chart(polar_range)

R2_comun = Manifold(2, r'\mathbb R^2')

R2_comun_chart.<x,y> = R2_comun.chart()

f = R2_polar.diff_map(R2_comun,

(r*cos(theta), r*sin(theta)),

name='f'

)

f.display()

[47]:
f : R2

polar −→ R2

(r,θ) 7−→ (x,y) = (r cos(θ) ,r sin(θ))
Primero lo hacemos en un punto específico.

[48]: p = R2_polar.point((1,pi/4),name='p')

fp = f(p)

p.coord(), fp.coord()

[48]: ((
1,

1
4

π

)
,

(
1
2

√
2,

1
2

√
2
))

Ahora calculamos la diferencial
[49]: dfp = f.differential(p)

display(dfp)

print(dfp)

dfp.parent()

dfp.matrix()

d f p

Generic morphism:

From: Tangent space at Point p on the 2-dimensional differentiable

manifold \mathbb R�2_\text{polar}

To: Tangent space at Point f(p) on the 2-dimensional

differentiable manifold \mathbb R�2

[49]: ( 1
2

√
2 −1

2

√
2

1
2

√
2 1

2

√
2

)
Hagamos lo mismo pero en un punto genérico.

[50]: r0,theta0 = var("r0 theta0")

assume(r0>0)

p = R2_polar.point((r0,theta0),name='p')

print(p)

display(p.coord())

dfp = f.differential(p)

dfp.matrix()

Point p on the 2-dimensional differentiable manifold

\mathbb R�2_\text{polar}

(r0,θ0)

[50]: (
cos(θ0) −r0 sin(θ0)
sin(θ0) r0 cos(θ0)

)
7. CAMPOS VECTORIALES

Un campo vectorial X en M asigna a cada p ∈ M un vector tangente Xp ∈ TpM de manera
“suave”. Si x1, . . . ,xn son coordenadas locales tenemos, asociados n campos coordenados
(locales)

∂

∂x1
,

∂

∂x2
, . . . ,

∂

∂xn

los cuales se corresponden con los campos constantes e1,e2, . . . ,en en Rn. Los campos coor-
denados son linealmente independientes en cada punto y por ende generan cada espacio
tangente en su dominio de definición: si v ∈ TpM,

v =
n

∑
i=1

vi
∂

∂xi

∣∣∣∣
p

Si f : M → R,

v( f ) =
n

∑
i=1

vi
∂ f
∂xi

∣∣∣∣
p
, en donde

∂ f
∂xi

=
d
dt

f (ci(t)).

Aquí ci(t) es la curva cuyas coordenadas son (x1(p), . . . ,xi(p)+t, . . . ,xn(p)). De esto último
se desprende que vi = v(xi).

Nuevamente hacemos un ejemplo en dimensión 4.

[51]: M = Manifold(4, 'M')

M_chart.<x,y,z,w> = M.chart()

R.<t> = manifolds.RealLine()

f = M.scalar_field(4*x*y + z^2 - cos(w*y))

Para acceder a los campos coordenados de una carta local tenemos el método .frame().

[52]: M_chart.frame()

[52]: (
M,

(
∂

∂x
,

∂

∂y
,

∂

∂ z
,

∂

∂w

))
Llamemos e0,e1,e2,e3 a estos campos coordenados y usemos esto para definir un nuevo

campo X .

[53]: e0,e1,e2,e3 = M_chart.frame()

X = x*e0 - e1 + z*w*e3; X

[53]:

Vector field on the 4-dimensional differentiable manifold M
Notemos que X aún no tiene un nombre en LATEX, pero esto lo podemos resolver rápida-

mente con .set_name().
[54]: X.set_name('X')

display(X)

X.display()

X
[54]:

X = x
∂

∂x
− ∂

∂y
+wz

∂

∂w
Ahora calculemos la derivada direccional X( f ).

[55]: display(f)

display(f.display())

display(X(f))

X(f).disp()

Scalar field on the 4-dimensional differentiable manifold M

M −→ R
(x,y,z,w) 7−→ 4xy+ z2 − cos(wy)

Scalar field on the 4-dimensional differentiable manifold M
[55]:

M −→ R
(x,y,z,w) 7−→ wyzsin(wy)+4xy−wsin(wy)−4x

También podemos usar el método X(f).expr() para acceder a la expresión simbólica
de X( f ), por si necesitamos usarla en otro lado.

[56]: display(X(f).expr())

print(X(f).expr())

wyzsin(wy)+4xy−wsin(wy)−4x
w*y*z*sin(w*y) + 4*x*y - w*sin(w*y) - 4*x

Lo siguiente también puede ser útil si en algún momento necesitamos reciclar alguna
expresión en LATEX.

[57]: print(latex(X(f).expr()))

w y z \sin(w y) + 4 \, x y - w \sin(w y) - 4 \, x

8. FLUJO DE CAMPOS VECTORIALES

Todo campo vectorial X ∈ X(M) se mueve a lo largo de unas curvas llamadas curvas
integrales o líneas de flujo cuya velocidad es exactamente el valor del campo en el punto
dado. Más precisamente, una curva integral de X por p ∈ M es una curva γp : I ⊂ R→ M
tal que γp(0) = p y γ ′p(t) = Xγp(t) para todo t ∈ I. Si pensamos ahora que t está fijo y p es
quien varía obtenemos una familia de difeomorfismos locales

ϕt : U ⊂ M → M, ϕt(p) = γp(t)

El conjunto {ϕt} forma un grupo (local) llamado el flujo de X ya que

ϕ0 = idM ϕt ◦ϕs = ϕt+s (ϕt)
−1 = ϕ−t

El flujo local de un campo X puede pensarse como un grupo de simetrías de X , pues

(ϕt)∗(X) = X

para todo t ∈ I.

9. CORCHETE DE CAMPOS VECTORIALES Y ÁLGEBRAS DE LIE

El conjunto X(M) de todos los campos vectoriales en M forma un un C∞(M)-módulo.
En efecto, ya que los campos vectoriales se pueden sumar y multiplicar por funciones dife-
renciables. El corchete de Lie en X(M) es una de las formas que tenemos para derivar un
campo en la dirección de otro. Si X = ∑ai

∂

∂xi
, Y = ∑b j

∂

∂x j
, definimos el corchete de Lie

entre X e Y como

[X ,Y ] = XY −Y X = ∑
i

∑
j

(
a j

∂bi

∂x j
−b j

∂ai

∂x j

)
∂

∂xi

Notemos que la definición de [X ,Y ] es independiente del sistema de coordenadas, pues
se tiene que aplicado una función f ∈C∞(M), vale

[X ,Y ]( f ) = X(Y ( f ))−Y (X( f )).

El espacio vectorial (sobre R) X(M) junto con el corchete de Lie forma un álgebra de
Lie real (de dimensión infinita), ya que se cumplen

[ , ] es R-bilineal.
Antisimetría: [X ,Y ] =−[Y,X ].
Identidad de Jacobi: [X , [Y,Z]]+ [Y, [Z,X ]]+ [Z, [X ,Y ]] = 0.

Además se satisface la regla de Leibniz: si f ∈C∞(M),

[X , fY ] = X( f )Y + f [X ,Y ]

El siguiente resultado nos dice que el corchete de Lie se puede usar para detectar cuando
dos campos son campos coordenadas (con respecto a alguna carta coordenada).

Teorema. Dos campos X e Y conmutan (i.e. [X ,Y ] = 0) si y sólo si sus flujos conmutan
ϕX

t ◦ϕY
s = ϕY

s ◦ϕX
t .

Corolario (Teorema de Clairaut). Los campos coordenados conmutan.

Demostración. En realidad esto es inmediato de la definición del corchete, pero hagámoslo
usando flujos a modo de ejemplo. Como este es un teorema local, es suficiente probarlo
para M = Rn y los campos coordenados ∂

∂xi
= ei. Aquí podemos calcular explícitamente

ϕ i
t (p) = tei + p y por ende ϕ i

t (ϕ
j

s (p)) = tei + se j + p = ϕ
j

s (ϕ
i
t (p)). □

Verifiquemos con Sage que los campos coordenados conmutan.

[58]: M = Manifold(3, 'M')

M_chart.<x,y,z> = M.chart()

M_frame = M_chart.frame()

M_frame

{X.bracket(Y).is_zero() for X in M_frame for Y in M_frame}

[58]:

{True}

En general dos campos cualesquiera no son conmutativos.

[59]: X, Y, Z = M_frame

X0, X1, X2 = (x^2 + y^2 + z^2+1)*X, (y^2 + sin(z))*Y, exp(-x*2)*Z

X0.set_name('X_0'), X1.set_name('X_1'), X2.set_name('X_2')

display(X0.bracket(X1).display())

display(X0.bracket(X2).display())

display(X1.bracket(X2).display())

[X0,X1] =
(
−2y3 −2ysin(z)

) ∂

∂x

[X0,X2] =−2ze(−2x) ∂

∂x
−2
(
x2 + y2 + z2 +1

)
e(−2x) ∂

∂ z

[X1,X2] =−cos(z)e(−2x) ∂

∂y
Los campos X0,X1,X2 son linealmente independientes en cada punto y por lo tanto for-

man una base de cada espacio tangente. Esto se conoce como un frame. Sage nos permite
definir frames que no necesariamente estén inducidos por un sistema de coordenadas. Esto
será muy importante cuando trabajemos con grupos de Lie.

[60]: M_other_frame = M.vector_frame('X', [X0, X1, X2])

M.frames()

[60]:

[(
M,

(
∂

∂x
,

∂

∂y
,

∂

∂ z

))
,(M,(X0,X1,X2))

]
Sin embargo, solo tenemos definido un sistema de coordenadas

[61]: M.atlas()

[61]:

[(M,(x,y,z))]

Sage chequea que el frame que queremos definir sea linealmente independiente en cada
punto (aunque no es bueno confiarse demasiado).

[62]: M.vector_frame('X', [X0, X1, X0+X1])

-------------------------------------------------------------------

ValueError Traceback (most recent call last)

/tmp/ipykernel_25603/2563714674.py in <module>

----> 1 M.vector_frame('X', [X0, X1, X0+X1])

/usr/lib/python3.9/site-packages/sage/manifolds/differentiable/

manifold.py in vector_frame(self, *args, **kwargs)

3152 "input matrix must be nonsingular"]

3153 if linked:

-> 3154 raise ValueError("the provided vector fields are not "

3155 "linearly independent")

3156 # Adding the newly generated changes of frame to the

ValueError: the provided vector fields are not linearly independent

10. GRUPOS DE LIE

Un grupo de Lie G es a la vez un grupo y una variedad diferenciable tal que las opera-
ciones

m : G×G → G i : G → G

(x,y) 7→ xy x 7→ x−1

son diferenciables. Algunos ejemplos importantes son:

El grupo lineal general GLn(R) de matrices invertibles n× n. Notar que GLn(R) =
det−1(R−{0}) es un grupo de Lie, ya que es un abierto en Mn×n(R)≃Rn2

. Muchos
(casi todos) ejemplos de grupos de Lie se pueden presentar como subgrupos del
grupo lineal general.
El grupo ortogonal On = {A ∈ GLn(R) : AAT = In} el cual es un subgrupo compacto
de GLn(R).
El grupo lineal especial SLn(R) = {A ∈ GLn(R) : detA = 1}.

10.1. Ejemplo: SL2(R). El grupo lineal especial SL2(R) es muy importante en muchas
áreas de la matemática. Si quisiéramos probar que SL2(R) es una variedad diferenciable
con lápiz y papel usaríamos el teorema de la función implícita para variedades (que tam-
bién nos da información sobre su topología). Para definir en Sage las coordenadas de una
matriz A ∈ SL2(R) lo haremos a mano escribiendo una de las entradas de A como función

de las otras tres. En primer lugar, escribimos A =

(
a b
c d

)
. Como ad − bc = 1, tenemos

que a ̸= 0 implica d = 1+bc
a y b ̸= 0 implica c = ad−1

b . Además SL2(R) = U ∪V con
U ≃ {(a,b,c) ∈ R3 : a ̸= 0} y V ≃ {(a,b,d) ∈ R3 : b ̸= 0}. Los símbolos ≃ los hemos
usado para indicar difeomorfismo (una función diferenciable biyectiva con inversa diferen-
ciable).

Veamos nuestro primer ejemplo de una variedad diferenciable que no podemos cubrir con
un solo sistema de coordenadas. En primer lugar, debemos definir los abiertos U y V usando
el método .open_subset() y luego indicamos que con estos dos abiertos ya tenemos todos
los puntos de SL2(R) usando .declare_union().

[63]: SL2 = Manifold(3, name=r"SL_2(\mathbb R)")

U = SL2.open_subset("U")

V = SL2.open_subset("V")

SL2.declare_union(U, V)

display(U.union(V))

display(SL2.open_cover_family())

SL2(R)

{{SL2(R)},{U,V}}
Ahora definimos las coordenadas en U y V como hicimos más arriba.

[64]: rest_a = lambda a00,a01,a10: a00!=0

rest_b = lambda b00,b01,b11: b01!=0

a.<a00,a01,a10> = U.chart(coord_restrictions=rest_a)

b.<b00,b01,b11> = V.chart(coord_restrictions=rest_b)

Vale la pena aclarar que el parámetro coord_restrictions= es usado precisamente
para indicar las coordenadas que queremos que sean no nulas en U y en V . En versiones
anteriores de SageMath esto se hacía con el método .add_restrictions() que ahora es
obsoleto. Ahora tenemos un atlas con dos sistemas de coordenadas:

[65]: SL2.atlas()

[65]:

[(U,(a00,a01,a10)) ,(V,(b00,b01,b11))]

Pero Sage aún piensa que U y V son abiertos disjuntos. Para resolver esto tenemos que

definir el cambio de coordenadas. Si en la matriz A =

(
a b
c d

)
tenemos que a ̸= 0 y b ̸= 0

tendremos que

a00(A) = a, a01(A) = b, a10(A) = c,

b00(A) = a, b01(A) = b, b11(A) = d.

Luego

b00 = a00, b01 = a01, b11 =
1+a01a10

a00
.

Para explicitar cómo son las coordenadas de b como función de las coordenadas de a usamos
a.transition_map(b, ...)

[66]: a_to_b = a.transition_map(b,

(a00, a01, (1+a01*a10)/a00),

intersection_name='W',

restrictions1= [a00!=0,a01!=0],

restrictions2 = [b00!=0,b01!=0]

)

a_to_b.display()

[66]: 
b00 = a00
b01 = a01

b11 = a01a10+1
a00

Para el otro cambio de coordenadas podemos usar el método inverse().

[67]: b_to_a = a_to_b.inverse()

b_to_a.display()

[67]: 
a00 = b00
a01 = b01

a10 = b00b11−1
b01

Ahora aparecen nuevos sistemas de coordenadas en nuestro atlas, porque en la intersec-
ción de los dominios de dos cartas coordenadas, tenemos definidas dos tipos de coordena-
das.

[68]: SL2.atlas()

[68]:

[(U,(a00,a01,a10)) ,(V,(b00,b01,b11)) ,(W,(a00,a01,a10)) ,(W,(b00,b01,b11))]

Veamos un ejemplo de un elemento de SL2 que no está en dicha intersección.

[69]: A = SL2.point((0,16,-7),chart=b)

display(A in U)

display(A in V)

False

True
Podemos armar un pequeño programa que nos dé la representación matricial de los ele-

mentos de SL2.
[70]: def SL2_to_mat(M):

if M in U:

M00,M01,M10 = M.coord(chart=a)

M11 = (1+M01*M10)/M00

else:

M00,M01,M11 = M.coord(chart=b)

M10 = (M00*M11-1)/M01

return matrix(2,2,[M00,M01,M10,M11])

[71]: A_mat = SL2_to_mat(A); A_mat

[71]: (
0 16

− 1
16 −7

)
Con la representación matricial podemos usar todos los métodos que Sage tiene definidos

para matrices. Por ejemplo:

[72]: display(A_mat.parent())

display(A_mat.det())

display(A_mat.characteristic_polynomial())

A_mat.inverse()

Mat2×2(Q)

1

x2 +7x+1

[72]: (
−7 −16

1
16 0

)
También podemos escribir un programa reciba matrices de determinante 1 y devuelva

puntos de SL2.

[73]: def mat_to_SL2(A):

"""

Atención: este programa no chequea si A.det(1) == 1,

tampoco chequea si los coeficientes de A son reales.

"""

if A[0,0] != 0:

return SL2.point((A[0,0],A[0,1],A[1,0]),chart=a)

else:

return SL2.point((A[0,0],A[0,1],A[1,1]),chart=b)

[74]: A_mat = matrix(2,2,[0,1,-1,2]); A_mat

[74]: (
0 1

−1 2

)

[75]: A = mat_to_SL2(A_mat)

display(A in U)

display(A in V)

A.coord(b)

False

True

[75]:

(0,1,2)

Ejercicio. Pensar qué pasaría si hacemos A.coord().

11. CAMPOS INVARIANTES A IZQUIERDA

Si bien, en general, los grupos de Lie no se pueden cubrir con una sola carta, siempre
dispondremos de un frame global de campos invariantes a izquierda. Si G es un grupo de
Lie, e ∈ G es elemento neutro y v ∈ TeG. El campo invariante a izquierda Xv inducido por v
se define como

Xv|g = dLg|ev =
d
dt

∣∣∣∣
0
g · c(t), g ∈ G,

en donde c(0) = e y c′(0) = v. El corchete de Lie de dos campos invariantes a izquierda es
nuevamente invariante a izquierda. Más aún,

g= {X ∈ X(G) : X es invariante a izquierda}

es una subálgebra de Lie de dimensión finita dimg= dimG de X(G), llamada el álgebra de
Lie de G. El álgebra de Lie es la versión infinitesimal de un grupo de Lie y todo grupo de
Lie queda determinado localmente por su álgebra de Lie.

Tratemos de definir un campo invariante a izquierda en SL2 (esta es una de esas cosas
que es más fácil de hacer en papel que en Sage). En primer lugar, notemos que el álgebra
de Lie sl2(R) de SL2(R) queda determinada por TISL2(R).

Ejercicio. La derivada en t = 0 de una curva de matrices de determinante 1 que en 0 pasa
por I tiene traza 0. Es decir,

TISLn(R) = {X ∈ Mn×n(R) : trA = 0}.

Tomemos, por ejemplo,

v = c′(0) =
(

0 1
0 0

)
, donde c(t) =

(
1 t
0 1

)
.

Sea X el campo invariante a izquierda tal que XI = v. Calculemos XA con A =

(
a b
c d

)
:

d
dt

∣∣∣∣
0
A · c(t) = d

dt

∣∣∣∣
0

(
a at +b
c ct +d

)
Para ver cuáles son las componentes de X tenemos que trabajar en coordenadas. Si A∈U ,

d
dt

∣∣
0(a,at +b,c) = (0,a,0).

Por otro lado, si A ∈V ,
d
dt

∣∣
0(a,at +b,ct +d) = (0,a,c) = (0,a,(ad −1)/b).

Para hacer todo esto en Sage, primero inicializamos un campo X

[76]: X = SL2.vector_field(name='X')

X.parent()

[76]:

X(SL2(R))
y luego indicamos cuáles son sus coeficientes según los campos coordenados:

[77]: X[a.frame(),:] = [0,a00,0]

X[b.frame(),:] = [0,b00,(b00*b11-1)/b01]

display(X.display(a))

display(X.display(b))

X = a00
∂

∂a01

X = b00
∂

∂b01
+

(
b00b11 −1

b01

)
∂

∂b11

Por la forma en que lo definimos, ya sabemos que X es invariante a izquierda. A modo
de ejemplo, verifiquemos con alguna matriz A que

XA = dLA|I(XI)

[78]: A = SL2.point((1,0,2),name='A')

display(SL2_to_mat(A))

display(A in U)

display(A in V)

(
1 0
2 1

)
True

False
Para calcular LA : SL2(R)→ SL2(R) necesitamos saber cuáles son las coordenadas de AB

para cualquier B ∈ SL2(R). En este caso tenemos una pequeña ventaja, ya que LA(U) ⊂U
y LA(V )⊂V .

Un punto genérico en U se puede obtener como sigue.

[79]: assume(a00!=0,b01!=0) #para que las coordenadas sean válidas

SL2.point(a[:])

SL2_to_mat(A), SL2_to_mat(SL2.point(a[:]))

[79]: ((
1 0
2 1

)
,

(
a00 a01

a10
a01a10+1

a00

))

Ahora podemos multiplicar normalmente las matrices

[80]: SL2_to_mat(A) * SL2_to_mat(SL2.point(a[:]))

[80]: (
a00 a01

2a00 +a10 2a01 +
a01a10+1

a00

)
y luego volver a SL2 con mat_to_SL2:

[81]: mat_to_SL2(SL2_to_mat(A) * SL2_to_mat(SL2.point(a[:]))).coord()

[81]:

(a00,a01,2a00 +a10)

Una forma fácil de obtener las coordenadas en el formato apropiado es usar print() y
luego copiarlas.

[82]: print(_)

(a00, a01, 2*a00 + a10)

Hacemos lo mismo con los puntos de V (teniendo cuidado de usar las coordenadas de la
carta b).

[83]: print(mat_to_SL2(SL2_to_mat(A) * \

SL2_to_mat(SL2.point(b[:],b))).coord(b))

(b00, b01, 2*b01 + b11)

Ya estamos en condiciones de definir el difeomorfismo LA ya que conocemos la expresión
en todos los sistemas de coordenadas.

[84]: LA = SL2.diff_map(SL2,{(a,a):(a00, a01, 2*a00 + a10),

(b,b):(b00, b01, 2*b01 + b11)},name="L_A")

LA.display()

[84]:
LA : SL2(R) −→ SL2(R)
on U : (a00,a01,a10) 7−→ (a00,a01,2a00 +a10)
on V : (b00,b01,b11) 7−→ (b00,b01,2b01 +b11)

Definimos la matriz identidad.
[85]: I2 = SL2.point((1,0,0),name='I_2')

display(SL2_to_mat(I2))

LA(I2)== A #chequeamos A * I2 = A

(
1 0
0 1

)
[85]:

True
Calculamos la diferencial dLA|I2 : TI2SL2(R)→ TASL2(R):

[86]: dLA = LA.differential(I2)

display(dLA.parent())

dLA.matrix()

Hom
(
TI2 SL2(R),TLA(I2) SL2(R)

)
[86]:  1 0 0

0 1 0
2 0 1


Y finalmente chequeamos que dLA|I2XI2 = XA.

[87]: dLA(X.at(I2)) == X.at(A)

[87]:

True

Ejercicio. 1) Escribir una función left_translation(A) que reciba un elemento A∈
SL2(R) y devuelva la traslación a izquierda LA : SL2(R)→ SL2(R).

2) Definir un frame invariante a izquierda en SL2.

12. TEORÍA DE LIE EN 2’

12.1. Teorema de Ado. Toda álgebra de Lie (real) de dimensión finita es isomorfa a una
subálgebra de Lie de gln(R) para algún n. Es decir, toda álgebra de Lie puede pensarse
como una subálgebra de matrices con respecto al corchete

[A,B] = AB−BA

12.2. Álgebras de Lie semisimples y solubles. Hay dos clases distinguidas de álgebras
de Lie de dimensión finita: las álgebras de Lie (semi)simples y las álgebras de Lie solubles.
Las álgebras de Lie semisimples son las “más sencillas”, están clasificadas y de esta clasifi-
cación (e.g. [Hel78]) sabemos que hay una cantidad finita en cada dimensión. Por otro lado
las álgebras de Lie solubles pueden presentarse como subálgebras de matrices triangulares,
aunque son bastante más complicadas de describir. No existe una clasificación completa y se
sabe que hay familias infinitas de álgebras solubles no isomorfas en una misma dimensión
(a partir de dimensión 3).

El llamado Teorema de descomposición de Levi nos dice que toda álgebra de Lie g tiene
una parte soluble y una parte semisimple que determinan completamente su estructura

g≃ gsol ⋊gss

12.3. La aplicación exponencial. Si G ⊂ GLn(R) es un grupo de Lie y g⊂ gln(R) es su
álgebra de Lie, la función exponencial nos da un difeomorfismo local entre un entorno del
origen en g y un entorno de la identidad en G.

exp : g→ G

A 7→ eA = In +A+
1
2!

A2 +
1
3!

A3 + · · ·

Una familia muy importante de grupos/álgebras de Lie solubles son los grupos/álgebras
de Lie nilpotentes. Si pensamos matricialmente un álgebra de Lie nilpotente, sus elementos
se pueden representar con matrices triangulares estrictas. Una ventaja de los grupos de Lie
solubles y, sobre todo, los nilpotentes (simplemente conexos) es que son mucho más fáciles
de implementar en Sage, ya se pueden cubrir con una sola carta.

Teorema. 1. Si G es soluble simplemente conexo, entonces G es difeomorfo a Rn con
n = dimG.

2. Si G es nilpotente simplemente conexo, entonces exp : g→ G es un difeomorfismo.
Más aún, la serie de eA es una suma finita para toda A ∈ g.

Sage también viene con álgebras de Lie

[88]: sl2 = lie_algebras.sl(QQ,2, representation='matrix')

print(sl2)

display(sl2.gens())

display(sl2.is_abelian())

display(sl2.is_semisimple())

display(sl2.is_solvable())

Special linear Lie algebra of rank 2 over Rational Field((
0 1
0 0

)
,

(
0 0
1 0

)
,

(
1 0
0 −1

))
False

True

False
El ejemplo más famoso de un álgebra de Lie nilpotente es el álgebra de Lie de Heisen-

berg, la cual tiene una base X ,Y,Z con el único corchete no trivial

[X ,Y ] = Z.

[89]: h = lie_algebras.Heisenberg(QQ, 1, representation="matrix");

print(h)

display(h.dimension())

X, Y, Z = h.basis()

display((X, Y, Z))

display(h.gens())

display(X.bracket(Y) == Z)

display(h.is_nilpotent())

display(h.is_semisimple())

Heisenberg algebra of rank 1 over Rational Field

3 0 1 0
0 0 0
0 0 0

 ,

 0 0 0
0 0 1
0 0 0

 ,

 0 0 1
0 0 0
0 0 0


 0 1 0

0 0 0
0 0 0

 ,

 0 0 0
0 0 1
0 0 0


True

Actas del XVI Congreso Dr. Antonio A. R. Monteiro (2021), 2023



GEOMETRÍA DE GRUPOS DE LIE USANDO SAGEMATH 95

True

False

13. MÉTRICAS RIEMANNIANAS

Una métrica riemanniana g, a veces también denotada por ⟨·, ·⟩, en una variedad diferen-
ciable M asigna a cada p ∈ M un producto escalar gp en TpM de manera diferenciable: en
coordenadas locales las funciones

gi j = g
(

∂

∂xi
,

∂

∂x j

)
son funciones diferenciables. Esto permite:

Calcular la norma de vectores tangentes (y de campos vectoriales): ∥v∥=
√

g(v,v).
Calcular ángulos entre vectores tangentes g(v,w) = ∥v∥∥w∥cosθ , en donde v,w ∈
TpM.
Calcular longitudes de curvas c : [a,b]→ M haciendo

L(c) =
∫ b

a
∥c′(t)∥dt.

Esto hace de M un espacio métrico y de hecho, la topología métrica inducida por la distancia

d(p,q) = ı́nf
c
{L(c) : c(a) = p, c(b) = q}

coincide con la topología de M.
El ejemplo más sencillo de variedad riemanniana es el espacio euclídeo (en cada punto

el producto escalar es el mismo)

[90]: %display plain

E2.<x,y> = EuclideanSpace(2)

E2.categories()

[90]: [Join of Category of smooth manifolds over Real Field with 53 bits

of precision and Category of connected manifolds over Real Field

with 53 bits of precision and Category of complete metric spaces,

Category of smooth manifolds over Real Field with 53 bits of

precision,

Category of connected manifolds over Real Field with 53 bits of

precision,

Category of connected topological spaces,

Category of differentiable manifolds over Real Field with 53 bits

of precision,

Category of manifolds over Real Field with 53 bits of precision,

Category of complete metric spaces,

Category of metric spaces,

Category of topological spaces,

Category of sets,

Category of sets with partial maps,

Category of objects]

Sage utiliza la notación tensorial para representar las métricas riemannianas.

[91]: %display latex

display(E2.default_chart())

g = E2.metric()

g.display()

(
E2,(x,y)

)
[91]:

g = dx⊗dx+dy⊗dy

En la métrica euclídea los campos coordenados canónicos forman una base ortonormal.

[92]: X, Y = E2.default_frame()

display(g(X, X).display())

display(g(X, Y).display())

display(g(Y, Y).display())

g(ex,ex) : E2 −→ R
(x,y) 7−→ 1

g(ex,ey) : E2 −→ R
(x,y) 7−→ 0

g(ey,ey) : E2 −→ R
(x,y) 7−→ 1

También podemos ver la representación matricial de la métrica (por medio de una matriz
simétrica definida positiva).

[93]: g[:]

[93]: (
1 0
0 1

)
Calculemos a modo de ejemplo la longitud de una elipse (con respecto a la métrica eu-

clídea).

[94]: R.<t> = manifolds.RealLine()

c = E2.curve((3*cos(t),2*sin(t)), (t,0,2*pi), name='c')

c.plot()

[94]:

Actas del XVI Congreso Dr. Antonio A. R. Monteiro (2021), 2023



GEOMETRÍA DE GRUPOS DE LIE USANDO SAGEMATH 97

Para ello integramos la norma del vector velocidad.

[95]: v = c.tangent_vector_field()

display(v.display())

print(v.at(R(t)))

display(v.at(R(t)).parent())

v.at(R(t)) in E2.tangent_space(c(t))

c′ =−3 sin(t)ex +2 cos(t)ey

Vector c' at Point on the Euclidean plane E�2

TNoneE2

[95]:

True
Para calcular la norma de c′(t), necesitamos evaluar la métrica en c(t), es decir usamos

el producto interno gc(t) en el espacio tangente Tc(t)E2.

[96]: gt = g.at(c(t))

vt = v.at(R(t))

gt(vt,vt)

[96]:

4 cos(t)2 +9 sin(t)2

Finalmente integramos.

[97]: integrate(sqrt(gt(vt,vt)),t,0,2*pi)

[97]:

∫ 2π

0

√
4 cos(t)2 +9 sin(t)2 dt

Si queremos una aproximación numérica podemos usar .n().

[98]: _.n()

[98]:

15,86543958929059

13.1. Ejemplo: el plano hiperbólico H2. Usaremos la presentación del semiplano supe-
rior. Es decir, la estructura diferenciable es la usual heredada de R2,

H2 = {(x,y) : y > 0}

y la métrica riemanniana dada en el punto (x,y) por

g(x,y) =
1
y2 (dx⊗dx+dy⊗dy).

[99]: H2 = Manifold(2, name=r'\mathbb H^2', structure='Riemannian')

H2_chart.<x,y> = H2.chart(r'x:(-oo,+oo) y:(0,+oo)')

H2_chart.coord_range()

[99]:

x : (−∞,+∞) ; y : (0,+∞)

Introducimos la métrica manualmente:

[100]: g = H2.metric()

g[0,0] = 1/y^2

g[1,1] = 1/y^2

g.display()

[100]:

g =
1
y2 dx⊗dx+

1
y2 dy⊗dy

Consideremos la siguiente curva en H2:

[101]: R.<t> = manifolds.RealLine()

gamma = H2.curve((1,t), (t,0,oo), name=r"\gamma")

v = gamma.tangent_vector_field()

gamma.plot()

[101]:

y calculemos la longitud de la curva γ(t), para t ∈ [1,8], con respecto a la métrica hiper-
bólica.

[102]: gt = g.at(gamma(t))

vt = v.at(R(t))

integrate(sqrt(gt(vt,vt)),t,1,8)

[102]:

3 log(2)

que aproximadamente es

[103]: _.n()

[103]:

2,07944154167984

La longitud euclídea esta curva es claramente 7. La longitud hiperbólica es menor que 7
porque los vectores euclídeos se ven más pequeños con la métrica hiperbólica en los puntos
que tienen coordenada y > 0. Si calculáramos la longitud de γ(t) con t ∈ [1

2 ,1] nos debería
dar más grande que 1

2 .

[104]: display(integrate(sqrt(gt(vt,vt)),t,1/2,1))

_.n()

log(2)

[104]:

2,07944154167984

14. LA CONEXIÓN DE LEVI-CIVITA

Asociada a una métrica riemanniana tenemos una conexión

∇ : X(M)×X(M)→ X(M), (X ,Y ) 7→ ∇XY

llamada conexión de Levi-Civita caracterizada por las propiedades:
∇ es R bilineal;
∇ es tensorial en la primera variable: ∇ f XY = f ∇XY , para toda f ∈C∞(M);
se satisface la regla de Leibniz: ∇X( fY ) = X( f )Y + f ∇XY , para toda f ∈C∞(M);
∇ es una conexión métrica: X(g(Y,Z)) = g(∇XY,Z)+g(Y,∇X Z) * ∇ es sin torsión:
∇XY −∇Y X = [X ,Y ].

Remarcamos muy rápidamente algunas similitudes y diferencias entre entre el corchete
de Lie y la conexión de Levi-Civita

Ambos se generalizan para poder derivar objetos más generales que campos (tenso-
res). La generalización del corchete de Lie se llama derivada de Lie y suele denotarse
por L , en tanto que la generalización de la conexión de Levi-Civita se denota por el
mismo símbolo.
Sin embargo, una detecta simetrías y la otra paralelismo. Más precisamente, si X es
un campo vectorial y T es un tensor,

• LX T = 0 dice que el flujo de X preserva T .
• ∇X T = 0 dice que T es paralelo en la dirección de X .

El corchete/derivada de Lie es independiente de la métrica.
La conexión de Levi-Civita está unívocamente determinada por la métrica.
La conexión de Levi-Civita permite derivar objetos que estén definidos solo a lo
largo de curvas. En particular, nos permite definir la aceleración de una curva (esto
no se puede hacer con la derivada de Lie).

14.1. Símbolos de Christoffel. Si M es una variedad riemanniana y x1, . . . ,xn es una carta
local, las propiedades anteriores nos dicen que para calcular ∇XY , es suficiente con saber
calcular ∇ ∂

∂xi

∂

∂x j
. Estos campos quedan determinados por los símbolos de Christoffel Γk

i j (en

la carta dada)

∇ ∂

∂xi

∂

∂x j
= ∑

k
Γ

k
i j

∂

∂xk
.

Sage nos permite calcular la conexión de Levi-Civita con g.connection() y nos mues-
tra los símbolos de Christoffel con g.connection().display(). Por ejemplo para el
plano hiperbólico:

[105]: nabla = g.connection()

nabla.display()

[105]:
Γ

x
xy = −1

y
Γ

x
yx = −1

y
Γ

y
xx = 1

y
Γ

y
yy = −1

y

Por defecto solo se muestran los símbolos de Christoffel no nulos, pero podemos pedir
que se muestren todos.

[106]: nabla.display(only_nonzero=False)

[106]:
Γx

xx = 0
Γ

x
xy = −1

y
Γ

x
yx = −1

y
Γ

x
yy = 0

Γ
y

xx = 1
y

Γ
y

xy = 0
Γ

y
yx = 0

Γ
y

yy = −1
y

También podemos acceder a los símbolos de Christoffel usando índices, notar que Γk
i j se

calcula con nabla[i,j,k].

[107]: display(nabla[1,0,0])

display(nabla[1,0,1])

display(nabla[1,1,1])

1
y

0

−1
y

15. GEODÉSICAS

Una curva γ(t) en una variedad riemanniana M se dice una geodésica si su velocidad es
paralela (a veces también se dice que tiene aceleración nula, pero esto se puede prestar a
confusión)

∇γ ′(t)γ
′(t) = 0.

SageManifolds todavía no tiene implementada la derivada covariante a lo largo de cur-
vas, pero podemos ayudarnos de los símbolos de Christoffel para estudiar la existencia de
geodésicas.

15.1. La ecuación geodésica. Si γ(t) es una geodésica tal que γ(0) = p y γ ′(0) = v ∈
TpM, se satisface

γ
′′
k +∑

i, j
Γ

k
i j γ

′
i γ

′
j = 0

En particular, siempre existen las geodésicas (para t pequeño) en cualquier dirección y
las geodésicas son las curvas que localmente minimizan la distancia.

15.2. Geodésicas en el plano hiperbólico. Se puede probar que las geodésicas en nues-
tra versión del plano hiperbólico son las semirrectas paralelas al eje y y los semicírcu-
los que intersectan perpendicularmente el eje x. El siguiente ejemplo fue tomado del fo-
ro de discusión de Sage y puede consultarse en https://ask.sagemath.org/question/48915/
check-if-a-curve-is-a-geodesic/.

[108]: H2 = Manifold(2, r'\mathbb H^2', structure='Riemannian')

X.<x,y> = H2.chart(r'x y:(0,+oo)')

g = H2.metric()

Actas del XVI Congreso Dr. Antonio A. R. Monteiro (2021), 2023

https://ask.sagemath.org/question/48915/check-if-a-curve-is-a-geodesic/
https://ask.sagemath.org/question/48915/check-if-a-curve-is-a-geodesic/


102 SILVIO REGGIANI

g[0, 0], g[1, 1] = 1/y^2, 1/y^2

g.display()

[108]:

g =
1
y2 dx⊗dx+

1
y2 dy⊗dy

Definamos la siguiente curva

[109]: R.<t> = manifolds.RealLine()

gamma = H2.curve([cos(t), sin(t)], (t, 0, pi), name=r'\gamma')

display(gamma.display())

v = gamma.tangent_vector_field()

display(v.display())

gamma.plot()

γ : (0,π) −→ H2

t 7−→ (x,y) = (cos(t) ,sin(t))

γ
′ =−sin(t)

∂

∂x
+ cos(t)

∂

∂y
[109]:

Nos ayudamos de los símbolos de Christoffel para estudiar la ecuación geodésica.

[110]: cc = g.christoffel_symbols()

a = gamma.domain().vector_field(dest_map=gamma)

for i in H2.irange():

a[i] = diff(v[i].expr(), t) \

+ sum(sum(cc[i, j, k](*X(gamma(t)))*v[j]\

.expr()*v[k].expr()

for j in H2.irange()) for k in H2.irange())

print(a)

a.display()

Vector field along the Real interval (0, pi) with values on the

2-dimensional Riemannian manifold \mathbb H�2

[110]:

cos(t)
∂

∂x
− cos(t)2

sin(t)
∂

∂y

Lo anterior nos dice que γ(t) no es una geodésica. Sin embargo, sabemos que la imagen
de γ es la imagen de una geodésica en el plano hiperbólico (lo que falla entonces es la para-
metrización). Tratemos de encontrar cuál es esa geodésica. Primero elijamos las condiciones
iniciales.

[111]: display(gamma(pi/2).coord())

v.at(R(pi/2)).plot() + gamma.plot(thickness=2)

(0,1)

[111]:

A continuación definimos una curva genérica γ(t) con coordenadas f1(t), f2(t). Las con-
diciones iniciales deberían ser:

f1(0) = 1 f ′1(0) = 0

f2(0) = 1 f ′2(0) =−1

[112]: f1 = function('f_1')(t)

f2 = function('f_2')(t)

Lo que estamos haciendo con esto es definir expresiones simbólicas que dependen de t.

[113]: print(type(f1))

<class 'sage.symbolic.expression.Expression'>

Las cuales luego pueden ser usadas para definir una curva genérica.

[114]: gamma = H2.curve([f1(t), f2(t)], t, name=r'\gamma')

v = gamma.tangent_vector_field()

v.display()

/tmp/ipykernel_25603/880341076.py:1: DeprecationWarning: Substitution

using function-call syntax and unnamed arguments is deprecated and

will be removed from a future release of Sage; you can use named

arguments instead, like EXPR(x=..., y=...)

See http://trac.sagemath.org/5930 for details.

gamma = H2.curve([f1(t), f2(t)], t, name=r'\gamma')

[114]:

γ
′ =

∂ f1

∂ t
∂

∂x
+

∂ f2

∂ t
∂

∂y

Aquí recibimos una amable advertencia de que el método que estamos usando para definir
γ es ahora obsoleto.

[115]: cc = g.christoffel_symbols()

a = gamma.domain().vector_field(dest_map=gamma)

for i in H2.irange():

a[i] = diff(v[i].expr(), t) \

+ sum(sum(cc[i, j, k](*X(gamma(t)))*v[j]\

.expr()*v[k].expr()

for j in H2.irange()) for k in H2.irange())

print(a)

a.display()

Vector field along the Real number line R with values on the

2-dimensional Riemannian manifold \mathbb H�2

[115]: (
−

2 ∂ f1
∂ t

∂ f2
∂ t

f2 (t)
+

∂ 2 f1

∂ t2

)
∂

∂x
+


(

∂ f1
∂ t

)2

f2 (t)
−

(
∂ f2
∂ t

)2

f2 (t)
+

∂ 2 f2

∂ t2

 ∂

∂y

Habría que resolver estas ecuaciones diferenciales, pero lo vamos a dejar acá.

16. ISOMETRÍAS

Una isometría entre dos variedades riemannianas es un difeomorfismo f : M → N que
preserva la métrica. Es decir,

d f |p : TpM → Tf (p)N
es una isometría lineal para cada p ∈ M. Se tiene que

I(M) = { f : M → M : f es una isometría}
forma un grupo con la composición, llamado el grupo de isometrías de M. El grupo I(M)
tiene una topología natural llamada topología compacto abierta y es bien conocido el si-
guiente resultado (ver por ejemplo [KN69]).

Teorema. I(M) es un grupo de Lie (con la topología compacto abierta) y si dimM = n,
entonces

dim I(M)≤ n(n+1)
2

.

Si se da la igualdad en el teorema anterior, entonces M resulta ser de una clase muy
especial: espacios de curvatura constante (veremos algo de esto en breve).

El álgebra de Lie del grupo de isometrías está dada por los campos de Killing: un campo
X ∈ X(M) se dice un campo de Killing si ∇X es antisimétrica con respecto a la métrica, es
decir:

⟨∇Y X ,Z⟩+ ⟨Y,∇ZX⟩= 0
para todos Y,Z ∈X(M). Hay una criterio muy sencillo (en teoría) para verificar si un campo
de X es un campo de Killing.

Teorema. X es un campo de Killing si y sólo si la derivada de Lie de la métrica en la
dirección de X es cero:

LX⟨·, ·⟩= 0.

Llevaría un buen tiempo definir qué es la derivada de Lie (es la extensión de la que habla-
mos antes del corchete de Lie para poder derivar todo tipo de tensores). Afortunadamente,
Sage sabe calcular derivadas de Lie. Sigamos explorando el ejemplo del plano hiperbólico

[116]: H2 = Manifold(2, r'\mathbb H^2', structure='Riemannian')

H2_chart.<x,y> = H2.chart(r'x y:(0,+oo)')

g = H2.metric()

g[0, 0], g[1, 1] = 1/y^2, 1/y^2

nabla = g.connection()

Calculemos las derivadas de Lie de la métrica en las direcciones de los campos coorde-
nados.

[117]: X, Y = H2.default_frame()

display(g.lie_derivative(X).disp())

display(g.lie_derivative(Y).disp())

0

− 2
y3 dx⊗dx− 2

y3 dy⊗dy

De lo anterior sigue que X = ∂/∂x es un campo de Killing. Verifiquemos que efectiva-
mente ∇X es un tensor antisimétrico

[118]: display(nabla(X))

nabla(X).display()

∇g
∂

∂x
[118]:

∇g
∂

∂x
=−1

y
∂

∂x
⊗dy+

1
y

∂

∂y
⊗dx

Otra forma de hacer esto es observando que la matriz de ∇X es antisimétrica, con respecto
al producto interno asociado a la métrica riemanniana (que en este caso es un múltiplo, que
varía punto a punto, del producto interno usual).

[119]: g[:], nabla(X)[:], nabla(Y)[:]

[119]: ((
1
y2 0
0 1

y2

)
,

(
0 −1

y
1
y 0

)
,

(
−1

y 0
0 −1

y

))

17. CURVATURA

El tensor de curvatura es uno de los objetos más importantes asociados a una variedad
riemanniana M, el cual captura casi toda la información geométrica de M. Usualmente de-
notado por R, es un tensor de tipo (1,3), lo cual informalmente podemos pensar como una
aplicación C∞(M)-multilineal: recibe tres campos y devuelve un campo

R : X(M)×X(M)×X(M)→ X(M)

y se define como
RX ,Y Z = ∇X ∇Y Z −∇Y ∇X Z −∇[X ,Y ]Z

En Sage podemos acceder al tensor de curvatura de la métrica g usando el método
g.riemann().

[120]: Riem = g.riemann()

print(Riem)

Riem.parent()

Tensor field Riem(g) of type (1,3) on the 2-dimensional Riemannian

manifold \mathbb H�2

[120]:

T (1,3) (H2)
En coordenadas locales, el tensor de curvatura tiene el siguiente aspecto.

[121]: %display plain

Riem.display()

[121]: Riem(g) = -1/y^2 ∂/∂x⊗dy⊗dx⊗dy + y^(-2) ∂/∂x⊗dy⊗dy⊗dx

+ y^(-2) ∂/∂y⊗dx⊗dx⊗dy - 1/y^2 ∂/∂y⊗dx⊗dy⊗dx

A veces conviene pensar al tensor de curvatura como un tensor de tipo (0,4), es decir,
recibe 4 campos y devuelve una función. Esto se hace contrayendo con la métrica

(X ,Y,Z,W ) 7→ ⟨RX ,Y Z,W ⟩
Sage sabe hacer esto con el método .down(g), pero hay que tener cuidado con los signos

(ya que no hay una convención universal sobre el orden en el cual ponemos las variables del
tensor de curvatura).

[122]: %display latex

Riem1 = Riem.down(g)

print(Riem1)

Tensor field of type (0,4) on the 2-dimensional Riemannian manifold

\mathbb H�2

[123]: Riem1.display()

[123]:

− 1
y4 dx⊗dy⊗dx⊗dy+

1
y4 dx⊗dy⊗dy⊗dx+

1
y4 dy⊗dx⊗dx⊗dy− 1

y4 dy⊗dx⊗dy⊗dx

Verifiquemos que ⟨R, ·⟩ se puede obtener con Riem.down(g). En primer lugar definamos
cuatro campos genéricos.

[124]: a0,a1,b0,b1,c0,c1,d0,d1 = var("a0,a1,b0,b1,c0,c1,d0,d1")

X1 = a0*X + a1*Y

X2 = b0*X + b1*Y

X3 = c0*X + c1*Y

X4 = d0*X + d1*Y

X1.display(), X2.display(), X3.display(), X4.display()

[124]: (
a0

∂

∂x
+a1

∂

∂y
,b0

∂

∂x
+b1

∂

∂y
,c0

∂

∂x
+ c1

∂

∂y
,d0

∂

∂x
+d1

∂

∂y

)
Observemos los siguiente, cuando decimos que R es un tensor, estamos pensando en

que es C∞(M)-multilineal (i.e., saca funciones diferenciables multiplicando). Por ende para
calcular RX ,Y Z en un punto p, solo necesitamos conocer Xp,Yp,Zp. En ese sentido es que
los campos X1,X2,X3,X4 que definimos más arriba son genéricos, ya que son combinaciones
lineales (sobre R) de un frame global.

Calculamos por un lado

[125]: comp_1 = Riem1(X1,X2,X3,X4).expr().expand()

comp_1

[125]:

−a1b0c1d0

y4 +
a0b1c1d0

y4 +
a1b0c0d1

y4 − a0b1c0d1

y4

y por el otro

[126]: comp_2 = g(nabla(nabla(X3)(X2))(X1) \

- nabla(nabla(X3)(X1))(X2), X4).expr().expand()

comp_2

[126]:
a1b0c1d0

y4 − a0b1c1d0

y4 − a1b0c0d1

y4 +
a0b1c0d1

y4

y verificamos que una es menos la otra.

[127]: comp_1 + comp_2

[127]:

0

18. CURVATURA SECCIONAL

La curvatura seccional es la versión de la curvatura gaussiana para variedades n dimen-
sionales. De hecho, cuando n = 2, nos dan la misma información. La curvatura secciona K
en un punto p de M es un valor que se asigna a cada subespacio bidimensional de TpM. Si
u,v es una base de V⊂ TpM

K(V) = K(u,v) =
⟨Ru,vv,u⟩

⟨u,u⟩⟨v,v⟩−⟨u,v⟩2 .

Se puede probar que está definición es independiente de la base elegida. Si n ≥ 3, K brinda
menos información que el tensor de curvatura, pero es un invariante más sencillo.

La curvatura seccional no viene por defecto en Sage, hay que definirla. Verifiquemos que
el plano hiperbólico tiene curvatura constante −1.

[128]: (Riem1(X1,X2,X1,X2) / (g(X1,X1) * g(X2,X2) - g(X1,X2)^2)).expr()

[128]:

−1
Y ya que estamos, definamos una función que calcule la curvatura seccional.

[129]: def curv_secc(u,v):

p = u.parent().base_point()

M = p.parent()

g = M.metric()

gp = g.at(p)

Riem = g.riemann().down(g).at(p)

return (Riem(u,v,u,v) / \

(gp(u,u)*gp(v,v) - gp(u,v)^2)).simplify_full()

p = H2.point(H2_chart[:])

u, v = H2_chart.frame().at(p)

curv_secc(u, v)

[129]:

−1

19. ESPACIOS DE CURVATURA CONSTANTE

Los espacios de curvatura constantes son aquellos en los que la curvatura seccional es
independiente del punto (y del plano elegido en el espacio). Hay 3 modelos simplemente
conexos de espacios de curvatura constante en cada dimensión n ≥ 2:

Rn con la métrica euclídea es un espacio de curvatura constante nula (también lla-
mado plano o flat);

Sn con la métrica redonda (heredada de Rn+1) es un espacio de curvatura constante
positiva;
Hn con la métrica hiperbólica es un espacio de curvatura constante negativa.

Para terminar el curso tratemos de probar el siguiente resultado con lo que aprendimos
hasta ahora.

Teorema. Sea M una variedad riemanniana simplemente conexa de dimensión 3 y curva-
tura constante. Entonces existe un grupo de Lie G con una métrica invariante a izquierda
isométrico a M.

19.1. Ejemplo: flat (con grupo de Lie abeliano). Es fácil ver a mano que el espacio
euclídeo R3 tiene curvatura nula. Más aún, probemos que cualquier métrica constante en R3

tiene curvatura cero (notar que R3 es un grupo de Lie abeliano y los campos coordenados
usuales son invariantes a izquierda).

[130]: M = Manifold(3, r'\mathbb R^3', structure='Riemannian')

M_chart.<x,y,z> = M.chart()

M_frame = M_chart.frame()

g = M.metric()

Definimos las siguientes variables simbólicas para construir una métrica constante arbi-
traria:

[131]: mu, nu, eta, alpha, beta, gamma \

= var("mu, nu, eta, alpha, beta, gamma")

g[0,0], g[1,1], g[2,2], g[0,1], g[0,2], g[1,2] \

= mu, nu, eta, alpha, beta, gamma

g[:] #esta matriz debe ser simétrica y definida positiva

[131]:  µ α β

α ν γ

β γ η


Veamos que los campos coordenados son invariantes a izquierda.

[132]: x0, y0, z0 = var('x0 y0 z0')

p = M.point((x0, y0, z0))

Lp = M.diffeomorphism(M, (x + x0, y + y0, z + z0), name='L_p')

Lp.display()

[132]:

Lp : R3 −→ R3

(x,y,z) 7−→ (x+ x0,y+ y0,z+ z0)

Aquí Lp es la traslación a izquierda en p = (x0,y0,z0) y su diferencial se identifica con la
transformación identidad.

[133]: zero = M.point((0, 0, 0))

dLp_0 = Lp.differential(zero)

dLp_0.matrix()

[133]:

 1 0 0
0 1 0
0 0 1


Como los campos coordenados tienen coordenadas constantes en el frame M_frame, si-

gue que dLp preserva los campos invariantes a izquierda. Además, como la métrica g tiene
coeficientes constantes en un frame invariante a izquierda, resulta una métrica invariante a
izquierda.

Por último, verifiquemos que la métrica es plana.

[134]: Riem = g.riemann()

%time Riem.display()

CPU times: user 1.17 ms, sys: 0 ns, total: 1.17 ms

Wall time: 1.19 ms

[134]:

Riem(g) = 0

19.2. Ejemplo: flat (con grupo de Lie soluble no abeliano). Es un poco más interesante
observar que existe un grupo de Lie no abeliano que admite una métrica plana. Considere-
mos el grupo euclídeo E(2) de movimientos rígidos del plano. Se puede ver que:

E(2) es un grupo de Lie de dimensión 3 y su estructura de grupo queda determinado
por el siguiente hecho: todo movimiento rígido es composición de una traslación Tv,
v ∈ R2 y una rotación Rt , alrededor del origen, en un ángulo t ∈ R:

(Tv ◦Rt)◦ (Tw ◦Rs) = (Tv ◦Rt ◦Tw ◦ (Rt)
−1)◦ (Rt ◦Rs)

= (Tv ◦TRt(w))◦Rt+s

= Tv+Rt(w) ◦Rt+s.

En coordenadas (v, t)= (x,y, t) el producto en el cubrimiento universal de E(2) queda

(x,y, t)(x′,y′, t ′) = (x+ x′ cos t − y′ sin t,y+ x′ sin t + y′ cos t, t + t ′).

[135]: M = Manifold(3, r'\tilde E(2)',structure='Riemannian')

M_chart.<x,y,t> = M.chart()

g = M.metric()

Definimos las traslaciones a izquierda usando la fórmula anterior

[136]: x0, y0, t0 = var("x0 y0 t0")

p = M.point((x0, y0, t0))

Lp = M.diffeomorphism(M, \

(x0+x*cos(t0)-y*sin(t0), \

y0+x*sin(t0)+y*cos(t0),t0+t), \

name='L_p')

Lp.display()

[136]:

Lp : Ẽ(2) −→ Ẽ(2)
(x,y, t) 7−→ (xcos(t0)− ysin(t0)+ x0,ycos(t0)+ xsin(t0)+ y0, t + t0)

y calculamos su diferencial

[137]: zero = M.point((0,0,0))

dLp_0 = Lp.differential(zero)

dLp_0.matrix()

[137]:  cos(t0) −sin(t0) 0
sin(t0) cos(t0) 0

0 0 1


Las columnas de la matriz dLp_0.matrix() nos dan los coeficientes del frame invariante

asociado, con respecto a M.default_frame().

[138]: X, Y, T = M.default_frame()

L0 = M.vector_field([cos(t),sin(t),0], name='L_0')

L1 = M.vector_field([-sin(t),cos(t),0], name='L_1')

L2 = M.vector_field([0,0,1], name='L_2')

M_left_inv_frame = M.vector_frame('L',(L0,L1,L2))

M.frames()

[138]: [(
Ẽ(2),

(
∂

∂x
,

∂

∂y
,

∂

∂ t

))
,
(
Ẽ(2),(L0,L1,L2)

)]
Verifiquemos que el corchete de campos invariantes a izquierda es invariante a izquierda.

[139]: display(L0.bracket(L1).display(M_left_inv_frame))

display(L0.bracket(L2).display(M_left_inv_frame))

display(L1.bracket(L2).display(M_left_inv_frame))

[L0,L1] = 0

[L0,L2] =−L1

[L1,L2] = L0

Para definir una métrica invariante a izquierda hay que usar un pequeño truco ya que a
.metric() no podemos pasar el el parámetro frame= pero a .tensor_field() sí. En-
tonces creamos un tensor g_aux con los mismo coeficientes que queremos para nuestra
métrica en un frame invariante a izquierda y luego definimos los coeficientes de g[i,j] =

g_aux[i,j] en el frame M.default_frame().

[140]: mu, nu, eta = var("mu, nu, eta")

assume(mu>0,nu>0,eta>0)

g_aux = M.tensor_field(0,2,\

[[mu,0,0],[0,nu,0],[0,0,eta]],\

frame=M_left_inv_frame,name='g_a')

g[:] = g_aux[:]

%display plain

g.display()

[140]: g = ((mu - nu)*cos(t)�2 + nu) dx⊗dx + (mu - nu)*cos(t)*sin(t) dx⊗dy

+ (mu - nu)*cos(t)*sin(t) dy⊗dx

+ (-(mu - nu)*cos(t)�2 + mu) dy⊗dy + eta dt⊗dt

La métrica es invariante a izquierda ya que tiene coeficientes constantes con respecto al
frame invariante a izquierda.

[141]: %display latex

%time g.display(M_left_inv_frame)

CPU times: user 1.08 s, sys: 9.68 ms, total: 1.09 s

Wall time: 900 ms

[141]:

g = µL0 ⊗L0 +νL1 ⊗L1 +ηL2 ⊗L2

Definimos el tensor de curvatura Riem de la métrica g.

[142]: %time Riem = g.riemann() #long time

CPU times: user 7.22 s, sys: 86.9 ms, total: 7.31 s

Wall time: 5.91 s

Hacer un Riem.display(M_left_inv_frame) lleva mucho tiempo, podemos acelerar
un poco la cosa con Parellelism().

[143]: Parallelism().set(nproc=8)

%display plain

%time Riem.display(M_left_inv_frame) #long time

CPU times: user 447 ms, sys: 114 ms, total: 561 ms

Wall time: 16.5 s

[143]: Riem(g) = 1/4*(mu�2 - 2*mu*nu + nu�2)/(eta*mu) L_0⊗L�1⊗L�0⊗L�1

- 1/4*(mu�2 - 2*mu*nu + nu�2)/(eta*mu) L_0⊗L�1⊗L�1⊗L�0

+ 1/4*(mu�2 + 2*mu*nu - 3*nu�2)/(mu*nu) L_0⊗L�2⊗L�0⊗L�2

- 1/4*(mu�2 + 2*mu*nu - 3*nu�2)/(mu*nu) L_0⊗L�2⊗L�2⊗L�0

- 1/4*(mu�2 - 2*mu*nu + nu�2)/(eta*nu) L_1⊗L�0⊗L�0⊗L�1

+ 1/4*(mu�2 - 2*mu*nu + nu�2)/(eta*nu) L_1⊗L�0⊗L�1⊗L�0

- 1/4*(3*mu�2 - 2*mu*nu - nu�2)/(mu*nu) L_1⊗L�2⊗L�1⊗L�2

+ 1/4*(3*mu�2 - 2*mu*nu - nu�2)/(mu*nu) L_1⊗L�2⊗L�2⊗L�1

- 1/4*(mu�2 + 2*mu*nu - 3*nu�2)/(eta*nu) L_2⊗L�0⊗L�0⊗L�2

+ 1/4*(mu�2 + 2*mu*nu - 3*nu�2)/(eta*nu) L_2⊗L�0⊗L�2⊗L�0

+ 1/4*(3*mu�2 - 2*mu*nu - nu�2)/(eta*mu) L_2⊗L�1⊗L�1⊗L�2

- 1/4*(3*mu�2 - 2*mu*nu - nu�2)/(eta*mu) L_2⊗L�1⊗L�2⊗L�1

Otra opción sería ir pidiéndole a Sage que haga cálculos más sencillos. Por ejemplo,

[144]: %display latex

Riem[1,0,1,0].factor()

[144]:

(
µ cos(t)2 −ν cos(t)2 +ν

)
(µ −ν)2

4ηµν

nos dice que para que la métrica sea plana necesitamos que µ = ν . Y podemos empezar
de nuevo. . .

[145]: g_aux = M.tensor_field(0,2,\

[[mu,0,0],[0,mu,0],[0,0,eta]],\

frame=M_left_inv_frame,name='g_a')

g[:] = g_aux[:]

%time Riem = g.riemann() #tarda mucho menos

CPU times: user 176 ms, sys: 295 ms, total: 471 ms

Wall time: 873 ms

Con esto ya hacemos que la métrica sea plana:

[146]: Riem.display()

[146]:

Riem(g) = 0
Esto también lo podemos comprobar viendo que la métrica invariante a izquierda coinci-

de con una métrica euclídea.
[147]: g.display()

g.display(M_left_inv_frame)

[147]:

g = µL0 ⊗L0 +µL1 ⊗L1 +ηL2 ⊗L2

19.3. Ejemplo: curvatura negativa. Para estudiar el caso del espacio hiperbólico nos
ayudaremos de ciertos resultados que son conocidos en la literatura. Consideremos el grupo
de Lie soluble G, cuya álgebra de Lie g admite una base e0,e1,e2 con los corchetes dados
por

[L0,L1] = 0, [L2,L0] = L0 [L2,L1] = L1

Se puede ver que el grupo de Lie simplemente conexo asociado a esta álgebra de Lie es
R3 con el producto

(x,y, t)(x′,y′, t ′) = (x+ etx′,y+ ety′, t + t ′).
Sabiendo esto, ya podemos implementarlo en Sage.

[148]: M = Manifold(3, 'M',structure='Riemannian')

M_chart.<x,y,t> = M.chart()

g = M.metric()

x0, y0, t0 = var("x0 y0 t0")

p = M.point((x0, y0, t0), name='p')

zero = M.point((0,0,0))

Lp = M.diffeomorphism(M, (x0+x*exp(t0),y0+y*exp(t0),t0+t), \

name='L_p')

Lp.display()

[148]:

Lp : M −→ M
(x,y, t) 7−→ (xet0 + x0,yet0 + y0, t + t0)

Ejercicio. Verificar que M es un grupo de Lie. Ayuda: Probar que el producto es aso-
ciativo definiendo otro punto arbitrario q, los difeomorfismos Lq, Lpq y chequeando que
Lp.post_compose(Lq) == Lpq. Finalmente, probar que existen inversos verificando que
Lp.inverse() == Lq con q = Lp.inverse()(zero).

Para calcular los campos invariantes a izquierda usamos el mismo truco que antes.

[149]: dLp_0 = Lp.differential(zero)

dLp_0.matrix()

[149]:  et0 0 0
0 et0 0
0 0 1


Y como antes definimos un frame invariante a izquierda.

[150]: X, Y, T = M.default_frame()

L0 = M.vector_field([exp(t),0,0], name='L_0')

L1 = M.vector_field([0,exp(t),0], name='L_1')

L2 = M.vector_field([0,0,1], name='L_2')

M_left_inv_frame = M.vector_frame('L',(L0,L1,L2))

M.frames()

[150]: [(
M,

(
∂

∂x
,

∂

∂y
,

∂

∂ t

))
,(M,(L0,L1,L2))

]
Chequeemos el corchete de Lie satisface las relaciones propuestas

[151]: display(L0.bracket(L1).display(M_left_inv_frame))

display(L2.bracket(L0).display(M_left_inv_frame))

display(L2.bracket(L1).display(M_left_inv_frame))

[L0,L1] = 0

[L2,L0] = L0

[L2,L1] = L1

Para hacer las cuentas más sencillas (y breves) consideramos métricas diagonales.

[152]: mu, nu, eta = var("mu, nu, eta")

assume(mu>0,nu>0,eta>0)

g_aux = M.tensor_field(0,2,[[mu,0,0],[0,nu,0],[0,0,eta]],

frame=M_left_inv_frame,name='g_a')

g[:] = g_aux[:]

display(g.display())

g.display(M_left_inv_frame)

g = µe(−2 t)dx⊗dx+νe(−2 t)dy⊗dy+ηdt ⊗dt

[152]:

g = µL0 ⊗L0 +νL1 ⊗L1 +ηL2 ⊗L2

[153]: %time Riem = g.riemann()

CPU times: user 347 ms, sys: 305 ms, total: 652 ms

Wall time: 1.2 s

Con esto ya podemos calcular la curvatura seccional y verificar que M es un espacio de
curvatura constante negativa.

[154]: u0,u1,u2,v0,v1,v2 = var("u0,u1,u2,v0,v1,v2")

u = u0 * L0.at(p) + u1 * L1.at(p) + u2 * L2.at(p)

v = v0 * L0.at(p) + v1 * L1.at(p) + v2 * L2.at(p)

curv_secc(u, v)

[154]:

− 1
η

19.4. Ejemplo: curvatura positiva. Este caso lo vamos a dejar como un ejercicio para
quien esté interesado. La idea es la siguiente. Consideramos la inmersión S3 ↪→ R4. La
métrica en S3 es la inducida de R4, vía la identificación

TpS3 = {p}⊥ = {v ∈ R4 : ⟨v, p⟩= 0}

Con esta métrica la esfera (de radio 1, centrada en el origen) es un espacio de curvatura
constate 1. Pero además la esfera es un grupo de Lie: si identificamos R4 con los cuaternio-
nes H, entonces

S3 = {p ∈H : |p|= 1}
es un subgrupo con la multiplicación (y resulta grupo de Lie porque la multiplicación en H
es diferenciable). En S3 hay un frame invariante a izquierda. En efecto, si X ,Y,Z ∈ X(H)
son los campos

Xp = ip Yp = jp Zp = kp

entonces X ,Y,Z son tangentes a la esfera y de hecho su restricción a S3 forma un frame
invariante a izquierda (porque multiplicar por i, j,k son isometrías en R4).

Ahora sólo falta implementar esto en Sage. Algunas ideas que pueden servir para hacer
esto.

[155]: #A partir de la version 9.3, la esfera ya viene precargada

S3 = manifolds.Sphere(3)

print(S3)

3-sphere S�3 of radius 1 smoothly embedded in the 4-dimensional

Euclidean space E�4

Por defecto S3 viene con coordenadas esféricas, pero también podríamos pedirle a Sage
que use las proyecciones estereográficas

[156]:

%display plain

iota = S3.embedding()

iota.display()

[156]: iota: S�3 � E�4

on A: (chi, theta, phi) 7→ (x1, x2, x3, x4) =

(cos(phi)*sin(chi)*sin(theta), sin(chi)*sin(phi)*sin(theta),

cos(theta)*sin(chi), cos(chi))

Podemos inducir la métrica de E4 a la esfera:
[157]: %display latex

gamma = S3.induced_metric()

print(gamma)

gamma.display()

Riemannian metric gamma on the 3-sphere S�3 of radius 1 smoothly

embedded in the 4-dimensional Euclidean space E�4

[157]:

γ = dχ ⊗dχ + sin(χ)2 dθ ⊗dθ + sin(χ)2 sin(θ)2 dφ ⊗dφ

Los campos X ,Y,Z en E4 son fáciles de definir. Por ejemplo

X =−x2
∂

∂x1
+ x1

∂

∂x2
− x4

∂

∂x3
+ x3

∂

∂x4

Solo faltaría definir sus restricciones a la esfera y probar que son invariantes a izquierda.

